<commit_before>SUMMARY = "Exalt is a network manager for the windows manager Enlightenment DR17"
HOMEPAGE = "http://watchwolf.fr/wiki/doku.php?id=exalt"

LICENSE = "LGPLv2.1"
LIC_FILES_CHKSUM = "file://COPYING;md5=fbc093901857fcd118f065f900982c24"

DEPENDS = "elementary vpnc wpa-supplicant ecore eet edbus"
RDEPENDS_${PN} = "vpnc wpa-supplicant"

PV = "0.9+svnr${SRCPV}"
SRCREV = "${EFL_SRCREV}"

inherit e
SRC_URI = "${E_SVN}/trunk;module=${SRCNAME};protocol=http;scmdata=keep"
S = "${WORKDIR}/${SRCNAME}"

EXTRA_OECONF = " \
--with-wpa_supplicant=${sbindir}/wpa_supplicant \
--with-vpnc=${sbindir}/vpnc \
--with-vpnc-disconnect=${sbindir}/vpnc-disconnect \
"

do_install_append() {
    # install dbus config file
    install -d ${D}${sysconfdir}/dbus-1/system.d/
    install -m 0644 data/daemon/dbus/exalt.conf ${D}${sysconfdir}/dbus-1/system.d/exalt.conf

    # install dbus service file
    install -d ${D}${datadir}/dbus-1/system-services/
    install -m 0644 org.e.Exalt.service ${D}${datadir}/dbus-1/system-services/
}

FILES_${PN} += "${datadir}/dbus-1"

<commit_msg>exalt: fix build for B!=S<commit_after>SUMMARY = "Exalt is a network manager for the windows manager Enlightenment DR17"
HOMEPAGE = "http://watchwolf.fr/wiki/doku.php?id=exalt"

LICENSE = "LGPLv2.1"
LIC_FILES_CHKSUM = "file://COPYING;md5=fbc093901857fcd118f065f900982c24"

DEPENDS = "elementary vpnc wpa-supplicant ecore eet edbus"
RDEPENDS_${PN} = "vpnc wpa-supplicant"

PV = "0.9+svnr${SRCPV}"
SRCREV = "${EFL_SRCREV}"

inherit e
SRC_URI = "${E_SVN}/trunk;module=${SRCNAME};protocol=http;scmdata=keep"
S = "${WORKDIR}/${SRCNAME}"

EXTRA_OECONF = " \
--with-wpa_supplicant=${sbindir}/wpa_supplicant \
--with-vpnc=${sbindir}/vpnc \
--with-vpnc-disconnect=${sbindir}/vpnc-disconnect \
"

do_install_append() {
    # install dbus config file
    install -d ${D}${sysconfdir}/dbus-1/system.d/
    install -m 0644 ${S}/data/daemon/dbus/exalt.conf ${D}${sysconfdir}/dbus-1/system.d/exalt.conf

    # install dbus service file
    install -d ${D}${datadir}/dbus-1/system-services/
    install -m 0644 ${B}/org.e.Exalt.service ${D}${datadir}/dbus-1/system-services/
}

FILES_${PN} += "${datadir}/dbus-1"


{'fraction_non_alphanumeric': 0.1780128794848206, 'fraction_numerical': 0.03955841766329347, 'mean_word_length': 4.62015503875969, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4807519', 'n_tokens_mistral': 983, 'n_tokens_neox': 919, 'n_words': 142}